

## Generating IMPEDANCE vs FREQUENCY Plots with APWIN

TN-1A

Impedance versus frequency plots can convey useful information in a number of situations. Perhaps the most common area is transducers, particularly loudspeakers, that have a very characteristic curve of impedance versus frequency. In fact, some speaker designers find this curve more meaningful than a conventional frequency response plot. Another area is electronic devices with a low input impedance. If they have reactive components in the input circuit, this will result in a variation of input impedance over frequency. This in turn will affect frequency response.

System One or System Two does not generate impedance versus frequency plots directly but here is a simple technique to produce them. Impedance of an unknown resistance can be derived by measuring voltage across a known resistance (Analog Generator Source Impedance) in series with the unknown resistance and using Ohm's law to solve for the unknown resistance. By connecting the unknown impedance across the GENERATOR OUTPUT terminals and measuring the voltage across this external load using GEN MON internal connection the actual impedance

of this external unknown impedance, in this case a speaker, can be calculated. Figure 1 shows the circuit. By sweeping the frequency of the generator, the impedance versus frequency graph can be produced.


If  $R_x$  is the unknown impedance  
 $R_s$  the generator source impedance  
 $V_{gen}$  the requested generator (open circuit) output voltage  
 $V_m$  the measured load voltage (GEN MON Reading)

then:

$$R_x = R_s \left( \frac{V_m}{V_{gen} - V_m} \right)$$

When this formula is applied to the measured voltage values, the result is an impedance versus frequency plot.

Figure 1 Simplified schematic of measurement path.



Shown below is a simple APBasic procedure that can be used with both System One and System Two to automate this process. It will perform the necessary computation, and update and display the results in the graph.

The unknown impedance should be lower than about 20k ohms for accurate results. As the unknown impedance be-

comes large in relation to the generator source impedance, the difference between  $V_{gen}$  and  $V_m$  becomes increasingly small to the point where the results become inaccurate. This would not be a limitation for common applications such as loudspeaker and 600 ohm circuit testing.

#### *APBasic Procedure code listing:*

```
Sub Main
'For System One connect the Analog Generator Channel A Output across DUT.
'For System Two connect the Analog Generator unbalanced Channel A Output across DUT.
'The default generator level, number of measurement points and graphic display settings
' can be defined by loading the "SpeakerZ.AT?" test and changing the default values and then
' saving the test with page 3 showing.

Rs = 600                                '600 Ohm source impedance.

Dim Vm() As Double

System = AP.Application.SysType

If System = 1 Then
    AP.File.OpenTest("SpeakerZ.AT1")      'Open System One test.
Else
    AP.File.OpenTest("SpeakerZ.AT2")      'Open System Two test.
End If

Vgen = InputBox$("Enter Generator Output Level", "Speaker Impedance Measurement", _
    Format(AP.Gen.Ampl("Vrms"), "##.000"))
If Vgen = "" Then End
AP.Gen.Ampl("Vrms") = Vgen

With AP.Prompt
    .Text = "Collecting DATA"
    .FontSize = 8                        'Set font size to 8 point.
    .Position(-1,-1,220,140)            'Set location and size.
    .Show                                'Show prompt.
End With

AP.Gen.Output = True                     'Turn Analog Generator Output ON.
AP.Sweep.Start                           'Collect data.
AP.Gen.Output = False                     'Turn Analog Generator Output OFF.

Vm = AP.Data.XferToArray(0,1,"V")        'Measurements of Generator Monitor Output Voltage.

'Apply formula to data to display Impedance in Ohms.
For Counter = 0 To AP.Data.ColSize(0,0) - 1
    Vm(Counter) = Rs * (Vm(Counter) / (Vgen - Vm(Counter)))
Next Counter

AP.Data.XferToArray(0,1,"V") = Vm        'Replace data in memory with Impedance calculations.

AP.Data.UpdateDisplay(0)
AP.Prompt.Hide                            'Hide prompt.
AP.Application.Page = 2
End Sub
```

For unusual situations where higher impedances are involved or an external amplifier is required, one can use an external impedance in series with the DUT.

If  **$R_x$**  is the unknown impedance  
 **$V_{amp}$**  the external amplifier output voltage  
 **$V_s$**  the voltage developed across the sense resistor  
 **$R_s$**  the sense resistor impedance

then:

$$R_x = \left( \frac{V_{amp}}{V_s / R_s} \right) - R_s$$


By dividing the voltage developed across the sense resistor by the sense resistor impedance we can determine the current flowing through the DUT. If we then divide the voltage across

the DUT and sense resistor by the voltage produced by the amplifier we can derive the combined impedance of the the DUT and sense resistor. Then all that is left to do is subtract the impedance of the sense resistor to determine the impedance of the DUT.

When selecting the current sense resistor it is important to select a precision non-inductive resistor. A good rule of thumb is to use a sense resistor with a value 1/100 of the device you are testing. The value of the sense resistor has to be large enough so that the amplitude measurements made across it are not noise limited. In addition, the value should not be large relative to the device being tested because of the potential loss in drive capability. Another problem with a high impedance sense resistor value is excessive power dissipation. In most cases a 1% 0.1 Ohm resistor will be good enough to provide acceptable results.

When this formula is applied to the measured voltage values, the result is an impedance versus frequency plot.

*Figure 2 Simplified schematic of measurement path when an external Amplifier and Sense Resistor are used.*



We have also prepared a procedure to automate this process. It runs the test to get the required data, performs the calculations, and displays the impedance versus frequency graph.

The program works best at mid-band frequencies but is usable from 10Hz to 10kHz.

Figures 3 and 4 on the following page are samples of an impedance graph of a multiway loudspeaker system made using the utilities described here.

#### *APBasic Procedure code listing:*

```
Sub Main
'Connect Analyzer Channel A Input to Amplifier Output (across DUT and sense resistor).
'Connect Analyzer Channel B Input across sense resistor that is in series with the speaker.
'The default generator level, number of measurement points and graphic display settings
' can be defined by loading the "Amp-SpkZ.AT?" test and changing the default values and then
' saving the test with page 3 showing.

Dim Vamp() As Double
Dim Vs() As Double

System = AP.Application.SysType

If System = 1 Then
    AP.File.OpenTest("Amp-SpkZ.AT1")           'Open System One test.
Else
    AP.File.OpenTest("Amp-SpkZ.AT2")           'Open System Two test.
End If

Rs = InputBox$("Enter current sense resistor value", "Speaker Impedance Measurement", ".1")

With AP.Prompt
    .Text = "Collecting DATA"
    .FontSize = 8                               'Set font size to 8 point.
    .Position(-1,-1,220,140)                   'Set location and size.
    .Show                                       'Show prompt.
End With


AP.Gen.Output = True
AP.Sweep.Start
AP.Gen.Output = False

Vamp = AP.Data.XferToArray(0,1,"V")           'Measurements of Amp Output Voltage.
Vs = AP.Data.XferToArray(0,2,"V")             'Measurements of Voltage developed across Sense Resistor.


'Apply formula to data to display Impedance in Ohms.
For Counter = 0 To AP.Data.ColSize(0,0) - 1
    Vamp(Counter) = (Vamp(Counter) / (Vs(Counter) / Rs)) - Rs
Next Counter

AP.Data.XferToArray(0,1,"V") = Vamp           'Replace data in memory with Impedance calculations.
AP.Data.XferToArray(0,2,"V") = Vamp

AP.Data.UpdateDisplay(0)
AP.Prompt.Hide                               'Hide prompt.
AP.Application.Page = 2
End Sub
```



*Figure 3 Plot of Impedance versus Frequency on a two-way loudspeaker system. The resonance point of each driver is clearly visible.*



*Figure 4 Plot of Impedance versus Frequency for a two-way speaker system. One trace shows the effect of mechanical loading. The magnitude and frequency of the resonance of the low frequency driver has been changed.*

*Here is a list of the files supplied with this TECH NOTE:*

FILENAME	PURPOSE
SPEAKERZ.APB	Procedure to automate impedance measurement using the Analog Generator and Analyzer only.
SPEAKERZ.AT1	System One test used with SPEAKERZ.APB
SPEAKERZ.AT2	System Two test used with SPEAKERZ.APB
AMP-SPKZ.APB	Procedure to automate impedance measurement using an external sense resistor.
AMP-SPKZ.AT1	System One test used with AMP-SPKZ.APB
AMP-SPKZ.AT2	System Two test used with AMP-SPKZ.APB

TECH NOTES are hints and information to assist Audio Precision test and measurement instrument users in performing specialized or unusual tests.

The contents of this TECH NOTE first appeared in the November 1992 issue of AUDIO.TST, the System One User's Newsletter.



Audio Precision, Inc.  
PO Box 2209  
Beaverton, Oregon 97075-3070  
U.S. Toll Free: 1-800-231-7350  
Tel: (503) 627-0832 Fax: (503) 641-8906  
email: [techsupport@audioprecision.com](mailto:techsupport@audioprecision.com)  
Web: [www.audioprecision.com](http://www.audioprecision.com)